* clean_enh_fuerza_trabajo.do
* This file applies some labels to the ENH datasets and cleans some of the variables.
* Cleans only the "fuerza_trabajo" files - Registro tipo 50
* Inputs: All files in C:/Users/JorgeEduardo/Dropbox/AA_Minimum_Wage/Data/Importable/ENH
* Outputs: All files in C:/Users/JorgeEduardo/Dropbox/AA_Minimum_Wage/Data/Clean/ENH

* Change log

/*	01/23/2015	JEP		Fix trabajo, change to trabajo_mas and avoid inconsitencies with trabajo_familia
*/		

clear all
version 13.1

* Project info
project, doinfo   
local master "`r(pdir)'"
local doname "`r(dofile)'" 



* Directory
cd "../../Data/Importable/ENH"

* Project calls
project, original("../../../Code/Clean/labels_enh_fuerza_trabajo.do")

* Registro tipo 50 - Fuerza trabajo

* Obtener una lista de variables comunes en todas las bases de datos
/* forvalues year=1996(1)2000 {
	forvalues trim=1(1)4 {
		use "`year'/TRIM `trim'/`year'_`trim'_fuerza_trabajo.dta" , clear
		unab vars: *
		if `year'==1996 & `trim'==1 local cv="`vars'"
		else local cv : list cv & vars
	}
}
di "`cv'" */

* tipo_registro id municipio segmento estrato sector seccion manzana vivienda hogar personas_hogar resultado_encuesta orden actividad trabajo_mas no_trabajo_empleo busco_ultima ha_buscado no_ha_buscado_t area fex


quietly {
	forvalues year=1996(1)2000 {
		forvalues trim=1(1)4 {
			project, uses("`year'/TRIM `trim'/`year'_`trim'_fuerza_trabajo.dta")
			use "`year'/TRIM `trim'/`year'_`trim'_fuerza_trabajo.dta" , clear
			do "../../../Code/Clean/labels_enh_fuerza_trabajo.do"
			drop if id==""
			drop if orden==.
			*** Base specific problems
			* Blanks in identifier in 1997
			if `year'==1997  replace id=subinstr(id," ","0",.)
			* Problema con un id en 1999 -4
			if `year'==1999 & `trim'==4 replace id="7300103630040010800201101" if id=="730010363004 010800201101"
			* Hay un hogar que tiene 43! personas en 1997-1. No sé si preparan la comida todos juntos. Mejor lo elimino
			drop if personas_hogar==43 
			
			* Work on common variables
			la val tipo_registro tipo
			**
			replace actividad = . if actividad==0
			la val actividad actividad			
			**
			cap confirm variable trabajo_mas
			if !_rc la val trabajo_mas sino
			**
			cap confirm string variable trabajo_familia
			if !_rc destring trabajo_familia, replace
			** 
			la val no_trabajo_empleo sino
			**
			la val busco_ultima sino
			**
			la val ha_buscado sino
			* Dummies de ingreso
			cap unab ding: d_ing*
			if !_rc {
				foreach var of varlist `ding' {
					replace `var'=0 if `var'==.
				}
			}
			**
			cap confirm string variable no_ha_buscado_t
			if !_rc destring no_ha_buscado_t, replace
			cap confirm numeric variable no_ha_buscado_t
			if !_rc replace no_ha_buscado_t=. if no_ha_buscado_t==99
			else replace no_ha_buscado_t="" if no_ha_buscado_t=="99"
			**
			cap confirm variable razon_ausente
			if !_rc replace razon_ausente = . if razon_ausente==9
			** Ingreso
			cap unab ing : ing_*
			if !_rc {
				foreach var of varlist `ing' {
					cap confirm numeric variable `var'
					if !_rc replace `var'=. if `var'==99
					else replace `var'="" if `var'=="99"
				}
			}
			* 
			cap confirm string variable ing_trabajo
			if !_rc destring ing_trabajo, replace
			cap confirm string variable ing_intereses
			if !_rc destring ing_intereses, replace
			*Codificación de área está diferente dentro de los años, ¡Qué sufrimiento!
			replace area=76 if area==761
			replace area=11 if area==111
			if `year'==1996 | `year'==1997 & `trim'<=2 {
				gen area961_972=area
				merge n:1 area961_972 using "../../xwalks/area.dta", update replace keep(1 3 4 5)
				drop if _merge==2
				drop _merge nombre
			}			
			la val area area
			* noi des, s
			* noi codebook, detail problems all
			* noi codebook, problems all
			compress 
			save "../../Clean/ENH/`year'/TRIM `trim'/`year'_`trim'_fuerza_trabajo.dta", replace
			project, creates("../../Clean/ENH/`year'/TRIM `trim'/`year'_`trim'_fuerza_trabajo.dta")
		}
	}
}
  